J. Linguistics **37** (2001), 339–353. © 2001 Cambridge University Press
DOI: 10.1017/S0022226701008830 Printed in the United Kingdom

NOTES AND DISCUSSION

German noun inflection revisited[1]

JAMES KILBURY

University of Düsseldorf

(Received 23 September 1999; revised 1 February 2001)

1. Introduction

Cahill & Gazdar (henceforth C&G) have presented their analysis of German noun inflection in issue 35.1 (1999) of this journal. As they emphasize (on page 4), the lexical knowledge representation language DATR they employ is theoretically neutral and can serve to encode descriptions set in entirely diverse theoretical frameworks, not just those that are theoretically close to theirs. The formalism is just as amenable to 'item and process' and 'word and paradigm' analyses as it is to the affixal 'item and arrangement' perspective. Moreover, distinct DATR theories (i.e., concrete descriptions) may differ greatly in their input (queries) and output (returned values) while they share a common structure reflecting the inheritance relations arising from the described phenomena.

In this paper I will present another analysis of German noun inflection, encoded in the same formalism but based on the theory of Minimalist Morphology developed by Wunderlich and his associates (Wunderlich & Fabri 1995, Wunderlich 1997a, 1999b). In his account German nouns are mapped into tree-based representations of their inflectional paradigms, whereas C&G map tuples of lexemes and inflectional categories (case and number) into individual inflected word forms. The major linguistic gain of my analysis is that the principal strength of Wunderlich's account, the formal description of relations WITHIN paradigms, is combined with the formal description of hierarchical relations BETWEEN paradigms, which is central for C&G but given little attention by Wunderlich.

2. German noun paradigms in Minimalist Morphology

Minimalist Morphology (henceforth MM), like many other theories of inflectional morphology, seeks to capture the notion of a paradigm. MM

[1] The work reported here was carried out within the Sonderforschungsbereich 282 *Theorie des Lexikons*, which is funded by the German Research Foundation. I wish to thank Petra Barg, Lynne Cahill, Gerald Gazdar, Dieter Wunderlich and an anonymous *JL* referee for comments on the original version of the paper.

formally represents paradigms as TREES in which (a) NODES are labelled with features specifying morphosyntactic information and (b) EDGES between nodes bear phonological transcriptions of affixes (cf. Wunderlich 1997a, 1999b; Steins 1998). MM handles the traditional notion of case within the framework of STRUCTURAL LINKING (cf. Wunderlich 1997b: 46–50) so that theta-roles are distinguished by the features [hr] 'there is a/no higher role' and [lr] 'there is a/no lower role'. For German, the linker specifications result in the following reduction of the traditional cases:

[ ]	nominative
[+hr]	accusative
[+hr, +lr]	dative
[+hr, +gen]	genitive

Note that MM assumes a principle of UNDERSPECIFICATION (cf. Fabri et al. 1996: 238, Wunderlich 1997a: 48–49) according to which the above representations are regarded as being distinct, although from the viewpoint of monotonic feature-based formalisms they are nondistinct and partially ordered by the subsumption relation. The representations become distinct under the convention that unspecified features be interpreted as negatively specified. Thus, [−hr] for nominative is distinct from [+hr,−lr] for accusative, and the latter is distinct from [+hr,+lr] for dative. Likewise, [ ] can be interpreted as [−pl] for singular and therefore distinct from [+pl] for plural.

A tree representing the paradigm of the noun *Arm* 'arm' is given in figure 1.



Figure 1
MM paradigm tree for *Arm*

The features appearing in node labels of the tree are minimally specified. Nodes inherit the features of all dominating nodes, so the leaf labelled [+lr] for dative plural *Arm-e-n* inherits [+pl] and [+m]. In contrast, special feature cooccurrence restrictions of German require that nodes specified as [+gen] or [+lr] also bear the feature [+hr]. The root is labelled for gender.

In the DATR representations here, constituent parts of trees can be referred to using a numerical code that amounts to a recipe for finding the parts. The first attribute of a path says what sort of thing you are looking for, e.g. a subtree, a branch, the affix labelling some edge, or the features labelling some node. The rest of the path tells you how to find this constituent: you start at the root node of the tree and stop once you get to a `0` in the DATR path. For other integers you count the edges emanating from the current node, take the one specified by the current integer, and go to the corresponding successor node. You then continue recursively with the remaining integers of the path until you finally reach a `0`. So `<tree 0>` designates the entire tree of figure 1, while `<tree 1 0>` and `<tree 2 0>` refer to the two subtrees immediately dominated by the root, and `<tree 2 1 0>` to the rightmost leaf. Likewise, `<branch 1 0>` designates the first subtree together with the edge leading to it, `<affix 1 0>` the label *-s* of the corresponding edge, and `<features 1 0>` the label `[+gen]` of the leaf.

The encoding of MM representations can be further simplified by omitting '+' before features since trees never contain explicit specifications with '−'. In a linear notation in which a colon separates edge labels from the subtree to which the edge leads, the tree of figure 1 appears as follows.[2]

```
([m] s:([gen]) e:([pl] n:([lr]))).
```

A concrete MM-based description of German noun inflection presupposes a series of design decisions:

- Homophonous affixes such as *-s* for both plural and genitive singular in *Klub* ('club') can be collapsed into single branches with disjunctive labels, i.e.,
  ```
  ([m] s:([gen or pl]))
  ```
  or kept distinct:
  ```
  ([m] s:([gen]) s:([pl])).
  ```
- Phonological representations of the affixes can be underlying, so that the tree for *Arm* also covers *Onkel* 'uncle' (with forms *Onkel* (NOM PL) and *Onkel-n* (DAT PL)) and *Schatten* 'shade' (invariant for all forms), or else surface-based.
- Assuming the latter of the last alternatives, the trees for *Onkel* and *Schatten* can be isomorphic to that for *Arm* but contain zero-allomorphs of the affixes, i.e.,
  ```
  ([m] ... 0:([pl] n:([lr])))
  ([m] ... 0:([pl] 0:([lr])))
  ```

[2] The transcriptions used in this paper are orthographic since, unlike C&G, I am essentially concerned only with morphotactics and not with the morphophonemics of German noun inflection. Thus, orthographic 'e' appears here for the vowel schwa.

or else can be structurally reduced to show the syncretism in the paradigms:

```
([m] ... n:([pl lr]))
     ([m] ... ).
```

Note that disjunctive labels on the last two trees, given the principle of underspecification in MM, would be both superfluous and senseless.

In each of the three choices just given, I have adopted the second possibility in this analysis since MM rejects unnatural feature descriptions like 'genitive or plural' for accidentally homophonous affixes and requires distinct affixes to show distinct surface realizations.

This is all that needs to be said here about the inflectional paradigms assumed in this paper since arguments for the analyses are presented elsewhere in the MM literature (cf. Wunderlich 1999b). Trees for a representative selection of German nouns are given below in Appendix B.[3]

3. Inflectional types and their hierarchical relations

Determining which MM paradigm should be assigned to a given noun is a prerequisite for the rest of an analysis of German noun inflection. It remains to specify (1) what distinct inflectional (i.e., paradigm) types appear, (2) what relations the types have to each other, and (3) how types can be assigned to individual nouns. Points (1) and (2) lie at the heart of the analysis by C&G and are captured by the nodes of their DATR theory and their inheritance relations; within MM they have been addressed by Wunderlich (1999a, b). Point (3) will be dealt with in this paper in section 4.

In the analysis here, which closely corresponds to that of C&G and my 1995 analysis cited by them (on page 24), I posit the inheritance structure between inflectional types depicted in figure 2 (see fn. 6 below for abbreviations).



Figure 2
The noun inflection hierarchy

[3] Computational linguists will inevitably wonder how such representations can be used for morphological parsing. A disarmingly simple answer is that the MM paradigm trees of Appendix B can be interpreted as DISCRIMINATION TREES or FINITE-STATE TRANSDUCERS (cf. Sproat 1992), so that well-established computational techniques can be brought to bear directly. But again, this goes beyond the scope of the present discussion.

Most differences between this and the earlier two analyses are insignificant. Classical loans like *Album* 'album' (with plural *Alben*) and highly irregular forms like *'Kaktus* 'cactus' (plural *Kak'teen*, with final stress) are ignored here. Some of the node names have been changed, and one new node has been introduced (see below).

The root of this hierarchy inherits from an additional node TOP, which specifies the syntax of the linear notation for trees. By default, features and branches are empty unless given a value at a lower node:[4]

```
TOP:
   <tree> == (<node-label>"<branches>")
   <node-label> == ["<features>"]
   <branch> == <edge-label>':' <tree>
   <edge-label> == "<affix>"
   <features> ==
   <branches> ==
   <auslaut> == Auslaut:<"<&&>" *>.[5]

   Auslaut:
   <$X e n *> == en-stem
   <$X e l *> == el-stem
   <$X e r *> == er-stem
   <$X e *> == e-stem
   <$X *> == non-schwa
   <$X> == <>.
```

The node Auslaut (the German term for final stem-shape) defines a finite state transducer that computes a morphophonemic type from the transcription of a stem. The first four axioms cover stems with a reduced final syllable (see below), while the fifth returns non-schwa for all other stems. The sixth axiom recursively shortens stem transcriptions until a final stem-shape can be assigned. The symbol '*' marks the end of a stem transcription.

---

[4] Note that in the following code the colon must be enclosed in single quotes when it refers to MM tree notation and does not have the reserved syntactic function of separating DATR nodes from paths.

[5] This last axiom uses the technical device of NODE CONVERSION (from the Düsseldorf QDATR implementation of DATR) with a predesignated path <&&> to convert any node name into a corresponding sequence of atoms. Since base forms of nouns are used as names of DATR nodes, this device, together with the following node Auslaut, determines the final shape of base forms, which is needed for the description (here, in contrast to C&G, orthographically based and highly simplified) of allomorphy in German noun inflection. Note that my rough and purely orthographically based definition of Auslaut does not account for nouns like *Gen* 'gene', *Gel* 'gel', *Ziel* 'goal' and *Meer* 'sea'.

A comprehensive treatment of German lexical entries would in my view derive both the orthographic and phonological representations of forms from a single underlying representation encoded in the DATR node name, but the details of this cannot be presented here. Clearly, it would be highly redundant to describe the orthography and pronunciation independently of each other.

Class NA nouns[6] like *Klub*, *Auto* and *Disco* are REGULAR (cf. Clahsen 1999)[7] but count for Wunderlich (1999a) as ATYPICAL since their plural forms, which end in *-s*, do not exhibit a reduced final syllable, unlike the nouns of all other classes. Paradigm roots are specified for gender, which is assumed here to be masculine by default, and nonfeminine nouns take the suffix *-s* in genitive singular:

```
NA:
   <> == TOP
   <features 0> == "<gnd>"
   <gnd> == m
   <branches 0> == IF:<EQ:<f "<gnd>">
                   THEN NULL
                   ELSE "<branch 1 0>">
                        "<branch 2 0>"
   <affix 1 0> == s
   <features 1 0> == gen
   <affix 2 0> == s
   <features 2 0> == pl.
```

This node definition states that assignments not explicitly made at node NA are otherwise inherited from TOP. In accord with the notation introduced above in section 2, the features labelling the root node of the paradigm tree can be referred to with the path `<features 0>`; the node label gives the gender of a noun, whatever it may be, but by default is specified here as masculine. The fourth axiom states that the branches emanating from the root, i.e., `<branches 0>`, consist of a plural branch `<branch 2 0>` which, depending on gender according to an IF-THEN-ELSE clause, may be preceded by a genitive branch `<branch 1 0>`. The last four axioms assign the labels of the nodes directly dominated by the root and of the edges leading to them.

The STRONG nouns like *Arm* of class NS follow NA in singular but add *-e* in plural unless the stem has a reduced final syllable, as in *Adler* 'eagle',

[6] In this discussion I use the following abbreviations for noun classes and the corresponding nodes of the inheritance hierarchy:

NA regular nouns (i.e., with *s*-plural)
NS strong nouns (i.e., with a schwa-syllable in plural)
NU strong nouns with umlaut in plural
NR strong nouns with *r*-plural
NM mixed nouns (with *n*-plural)
NWN weak nouns without *-s* in gen sg
NWS weak nouns with *-s* in gen sg

[7] Clahsen (1999: 994–995) explicitly identifies REGULAR or DEFAULT inflection with PRODUCTIVE processes. In DATR, however, defaults rest on a FORMAL notion of inheritance from a higher node in a network, and no special device is reserved to capture the COGNITIVE notion of productivity.

Laken 'sheet' and *Gebirge* 'mountain range', whose singular and plural forms are identical. Except for nouns like *Schatten* 'shadow' with a stem ending in *-en*, *-n* is added to the plural forms in dative plural. The dative leaf in the paradigm for *Arm* inherits the specification `[pl]` from the plural node, while *Adler* has no plural node and thus requires `[pl lr]` on the leaf:

```
NS:
   <> == NA
   <branch 2 0> == IF:<EQ:<en-stem "<auslaut>">
                   THEN NULL
                   ELSE IF:< EQ:<non-schwa
                                  "<auslaut>">
                        THEN NA
                        ELSE <branch 2 1 0>>>
   <affix 2 0> == e
   <branches 2 0> == <branch 2 1 0>
   <affix 2 1 0> == n
   <features 2 1 0> = =IF:<EQ:<non-schwa
                                "<auslaut>">
                       THEN lr
                       ELSE pl lr >.
```

The nouns of class NU like *Arzt* 'physician', *Floss* 'raft' and *Hand* 'hand' behave just like those of class NS but show umlaut in plural, giving *Ärzte*, *Flösse* and *Hände*, respectively (cf. C&G 1999: 10 ff.). The feature `[uml_pl]` merely provides a lexical marking for these forms.

```
NU:
   <> == NS
   <features 0> == NS uml_pl.
```

Finally for the strong nouns, those of class NR like *Mann* 'man' and *Buch* 'book' (all nonfeminine), mark the plural with both umlaut and a suffix *-r*; dative plural *-n* in *Männ-er-n* is inherited from NS via NU:

```
NR:
   <> == NU
   <affix 2 0> == r.
```

All other noun classes show a suffix *-n* in plural, and dative plural is not distinguished. The singular forms of the MIXED nouns in NM like *Staat* 'state', *Hemd* 'shirt' and *Zeit* 'time' (note that the class of *Zeit* will be handled differently below), are like those of NA:

```
NM:
   <> == NA
   <affix 2 0> == n.
```

With a single exception, the remaining WEAK nouns are masculine. Those in NWN, like *Bär* 'bear' and *Hase* 'hare', have a suffix *-n* in the singular nonnominative forms:

```
NWN:
   <> == NM
   <affix 1 0> == n
   <features 1 0> == hr.
```

The remaining class NMS, with *Fels* 'rock' and *Name* 'name' as well as the only neuter *Herz* 'heart', inherits from NMN but further adds the suffix *-s* in genitive singular, giving *Name-n-s*:

```
NWS:
   <> == NWN
   <branches 1 0> == "NA:<branch 1 0>".
```

In contrast to my earlier analysis (Kilbury 1995), I have introduced a distinction between the types NWN and NWS here. Masculines with stems ending in *-e* inflect according to their animacy, animates according to NMN, and inanimates to NMS (cf. section 4 below). However, some nouns such as *Planet* 'planet' and *Hydrant* 'hydrant', with stem-final consonants and final stress, have genitive singular forms *Planeten* and *Hydranten*, respectively, without *-s*, although both are inanimate. So weak animates have no *-s* in genitive singular, but not all the weak nouns without *-s* are animate (cf. C&G 1999: 20, Wunderlich 1999b).

DATR representations of lexical entries for all examples covered are given in Appendix A below. The dump produced by querying each lexical node with the path `<tree 0>` to obtain the corresponding paradigm tree appears as Appendix B.

4. Assignment of inflectional types to lexemes

Now that we have seen what Minimalist Morphology paradigms are assigned to German nouns, what inflectional (i.e., paradigm) types are distinguished and what hierarchical relations these types have to each other, it remains to be shown how inflectional types are assigned to the nouns. In the description above, as in that of C&G, DATR nodes representing the types are directly assigned in the lexical entries. This course is sensible and undoubtedly would be adopted in any straightforward DATR encoding. MM places great value on the claim, however, that 'the membership in an inflectional subclass should not be arbitrarily assigned, but rather follow from features that can be memorized' (Wunderlich 1999b: 7; cf. also Wunderlich & Fabri 1995, Steins 1998). This directly contradicts my analysis, but I will now show that the contradiction is only apparent and can be resolved.

Features have so far played only a minor role here. The feature [uml_pl] marks root nodes in the paradigms of class NU, but it has not been utilized to subsume NS and NU under a single class. Gender distinguishes the paradigms with respect to genitive singular forms, but it would be possible, for example, to differentiate class NA into NAf and NAnf for feminine and nonfeminine nouns, respectively, so that the inflection of all nouns would be determined solely by a single atomic designation of inflectional type.

In what follows I present an analysis that is close to that of Wunderlich (1999a, b) and which conforms to the MM requirement of nonarbitrary class assignment. The following features, some of which have already appeared as node labels, will be used:

nt	typical noun
f	feminine
m	masculine
schwa	stem-final *-e* (schwa)
inan	inanimate
RFS_pl	reduced final syllable in plural
uml_pl	umlaut in plural
r_pl	suffix *-r* in plural
n_pl	suffix *-n* in plural
n_obl	suffix *-n* in singular nonnominative (i.e., oblique)
ns_gen	*-n-s* in genitive singular

I shall assume that these features, most of which stem directly from Wunderlich (1999b), meet the MM requirement that they can be memorized. Table 1 shows the distribution I postulate for these features in lexical entries of German nouns.

Specified features are distinguished in the table as being distinctive ('+'), redundant ('×') (i.e., predictable from other specifications), or nondistinctive ('•') (i.e., irrelevant). Some of the gender specifications are nondistinctive but would become distinctive if we were to introduce the types NAf and NAnf as suggested above. Otherwise, *Klub*, *Auto* and *Disco* are unspecified and thus are assigned to class NA.

Among the nouns distinctively marked [nt], inanimate masculines in *-e* like *Name* (with the single exception of *Käse* 'cheese') go to NWS (and redundantly acquire the features of the latter), while animates in *-e* like *Hase* go to NWN, feminines like *Zeit* and *Farbe* to NM, and others like *Arm* and *Haar* to NS. Note that all weak masculines with consonantal stems like *Bär*, as well as some inanimates like *Planet*, are in NWN, while other inanimates like *Fels* go to NWS; thus the feature [ns_gen] is distinctive (i.e., unpredictable) only for weak inanimate masculines with consonantal stems.

The remaining classes of typical nouns exhibit hierarchical IMPLICATIONAL PATTERNS (cf. Bailey 1973) in their feature specifications. Nouns like *Mann* and *Buch*, distinctively specified [r_pl], are redundantly [uml_pl] while

		nt	f	m	schwa	inan	RFS_ pl	uml_ pl	r_ pl	n_ pl	n_ obl	ns_ gen
NA	Klub			•								
NA	Auto											
NA	Disco		•									
NS	Arm	+		•								
NS	Haar	+										
NM	Zeit	+	+							×		
NM	Farbe	+	+		•					×		
NWN	Hase	+		+	+					×	×	
NWS	Name	+		+	+	+				×	×	×
NS	Drangsal	×	•				+					
NU	Arzt	×		•			×	+				
NU	Floss	×					×	+				
NU	Hand	×	•				×	+				
NR	Mann	×		•			×	×	+			
NR	Buch	×					×	×	+			
NM	Staat	×		•						+		
NM	Hemd	×								+		
NWN	Bär	×		×						×	+	
NWN	Planet	×		×		•				×	+	
NWS	Fels	×		×		×				×	×	+

Table 1
Distribution of distinctive features in lexical entries

those like *Arzt*, *Floss* and *Hand*, distinctively specified `[uml_pl]`, are redundantly `[RFS_pl]`. Only highly exceptional forms like *Drangsal* 'hardship', which are typical and feminine but take neither *-n* nor umlaut in plural, must be distinctively specified `[RFS_pl]`. An analogous implicational pattern can be seen in *Fels*, *Bär* and *Staat* with the features `[ns_gen]`, `[n_obl]` and `[n_pl]`.

Finally, figure 3 shows the distribution of distinctive features and lexemes in the noun inflection hierarchy.

What all of this shows is that the lexical assignment of what appear to be arbitrary and diacritic classes is harmless in my analysis. Given table 1 we can construct corresponding lexical entries in DATR using a single class node for nouns and then COMPUTE the inflection types given in figures 2 and 3.[8] The

[8] This problem is being investigated by Wiebke Petersen for her doctoral dissertation at the University of Düsseldorf. See Ganter & Wille (1996) for related work.



Figure 3
Distribution of distinctive features and lexemes in the noun inflection hierarchy

computation amounts to a COMPILATION of the minimalist entries into a computationally more efficient form through a kind of PARTIAL EXECUTION (cf. Pereira & Shieber 1987: 98 ff.). This can easily be implemented in DATR, but it need not be shown now just how it would be done. The important point is that the inflectional class of a noun can be computed from the features of its lexical entry; the class, which may be viewed as an ALIAS for the corresponding bundle of features, then determines the assignment of a paradigm according to the inflectional hierarchy.

REFERENCES

Bailey, Charles-James N. (1973). *Variation and linguistic theory*. Arlington, VA: Center for Applied Linguistics.

Cahill, Lynne & Gazdar, Gerald (1999). German noun inflection. *Journal of Linguistics* **35.1**. 1–42.

Clahsen, Harald (1999). Lexical entries and rules of language: a multidisciplinary study of German inflection. *Behavioral and Brain Sciences* **22**. 991–1013.

Fabri, Ray, Urbas, Martina & Walther, Markus (1996). A computational model of minimalist morphology. *Theoretical Linguistics* **22**. 231–279.
Ganter, Bernhard & Wille, Rudolf (1996). *Formale Begriffsanalyse: mathematische Grundlagen*. Berlin: Springer.
Kilbury, James (1995). German noun inflection. Paper presented to the 3rd ESRC Workshop *Frontiers of Research in Morphology*, Brighton, June 1995.
Pereira, Fernando C. N. & Shieber, Stuart M. (1987). *Prolog and natural language analysis*. Stanford, CA: CSLI.
Sproat, Richard (1992). *Morphology and computation*. Cambridge, MA: MIT Press.
Steins, Carsten (1998). Against arbitrary features in inflection: Old English declension classes. In Kehrein, Wolfgang & Wiese, Richard (eds.), *Phonology and morphology of the Germanic languages*. Tübingen: Niemeyer. 241–265.
Wunderlich, Dieter (1997a). Der unterspezifizierte Artikel. In Dürscheid, Christa, Ramers, Karl Heinz & Schwarz, Monika (eds.), *Sprache im Fokus*. Tübingen: Niemeyer. 47–55.
Wunderlich, Dieter (1997b). Cause and the structure of verbs. *Linguistic Inquiry* **28**. 27–68.
Wunderlich, Dieter (1999a). German noun plural reconsidered. *Behavioral and Brain Sciences* **22**. 1044–1045.
Wunderlich, Dieter (1999b). German noun plural reconsidered. Ms., University of Düsseldorf. (Extended version of 1999a available under http://web.phil-fak.uni-duesseldorf.de/~wdl/.)
Wunderlich, Dieter & Fabri, Ray (1995). Minimalist morphology: an approach to inflection. *Zeitschrift für Sprachwissenschaft* **14**. 236–294.

Author's address: Institut für Sprache und Information,
Heinrich-Heine-Universität Düsseldorf,
Universitätsstr. 1,
D-40225 Düsseldorf,
Germany.
E-mail: kilbury@ling.uni-duesseldorf.de

APPENDIX A

A representative list of example lexical entries

```
% regular but atypical nouns (i.e. with s-plural)
Klub:       <> == NA.
Auto:       <> == NA   <gnd>  ==   .% neuter as [-m -f]
Disco:      <> == NA   <gnd>  ==  f.

% strong nouns (i.e. with a schwa-syllable in plural)
Arm:        <> == NS.
Adler:      <> == NS.
Schatten:   <> == NS.
Haar:       <> == NS   <gnd>  ==   .
Lager:      <> == NS   <gnd>  ==   .
Laken:      <> == NS   <gnd>  ==   .
Gebirge:    <> == NS   <gnd>  ==   .
Drangsal:   <> == NS   <gnd>  ==  f.

% strong nouns with umlaut in plural
Arzt:       <> == NU.
Acker:      <> == NU.
Garten:     <> == NU.
Floss:      <> == NU   <gnd>  ==   .
Kloster:    <> == NU   <gnd>  ==   .
Hand:       <> == NU   <gnd>  ==  f.
Mutter:     <> == NU   <gnd>  ==  f.

% strong nouns with r-plural (no feminines)
Mann:       <> == NR.
Buch:       <> == NR   <gnd>  ==   .

% mixed nouns (with n-plural)
Staat:      <> == NM.
Hemd:       <> == NM   <gnd>  ==   .
Zeit:       <> == NM   <gnd>  ==  f.
Farbe:      <> == NM   <gnd>  ==  f.
Sieben:     <> == NM   <gnd>  ==  f.

% weak nouns without -s in gen sg (masculine only)
Baer:       <> == NWN.
Hase:       <> == NWN.

% weak nouns with -s in gen sg
Fels:       <> == NWS.
Name:       <> == NWS.
Herz:       <> == NWS   <gnd>  ==   .% the only
                                    % non-masculine
                                    % of this class
```

APPENDIX B

DATR output of MM paradigm trees for the example lexical entries

```
Klub:<tree 0> =       ( [ m ] s: ( [ gen ] ) s: ( [ pl ] ) ).
Auto:<tree 0> =       ( [ ] s: ( [ gen ]) s: ( [ pl ] ) ).
Disco:<tree 0> =
                      ( [ f ] s: ( [pl] ) ).
Arm:<tree 0>=         ( [ m ] s: ( [ gen ] )
                          e: ( [ pl ] n: ( [ lr ] ) ) ).
Adler:<tree 0> =      ( [ m ] s: ( [ gen ] )
                          n: ( [ pl lr ] ) ).
Schatten:<tree 0> =   ( [ m ] s: ( [ gen ] ) ).
Haar:<tree 0> =       ( [ ] s: ( [ gen ] )
                        e: ( [ pl ] n: ( [ lr ] ) ) ).
Lager:<tree 0> =      ( [ ] s: ( [ gen ] )
                        n: ( [ pl lr ] ) ).
Laken:<tree 0> =      ( [ ] s: ( [ gen ] ) ).
Gebirge:<tree 0> =    ( [ ] s: ( [ gen ] )
                        n: ( [ pl lr ] ) ).
Drangsal:<tree 0> =   ( [ f ] e: ( [pl ]
                          n: ( [ lr ] ) ) ).
Artz:<tree 0> =       ( [ m uml_pl ] s: ( [ gen ] )
                                e: ( [ pl ]
                                     n: ( [ lr ] ) ) ).
Acker:<tree 0> =      ( [ m uml_pl ] s:( [ gen ] )
                                n: ( [ pl lr ] ) ).
Garten:<tree 0> =     ( [ m uml_pl ]s:( [ gen ] ) ).
Floss:<tree 0> =      ( [ uml_pl ] s: ( [ gen ] )
                              e: ( [ pl ]
                                   n: ( [ lr ] ) ) ).
Kloster:<tree 0> =    ( [ uml_pl ] s:( [ gen ] )
                              n: ( [ pl lr ] ) ).
Hand:<tree 0> =       ( [ f uml_pl ] e:( [ pl ]
                                n: ( [ lr ] ) ) ).
Mutter:<tree 0> =     ( [ f uml_pl ]n:( [ pl lr ] ) ).
Mann:<tree 0> =       ( [ m uml_pl ] s: ( [ gen ] )
                                r: ( [ pl ]
                                     n: ( [ lr ] ) ) ).
Buch:<tree 0> =       ( [ uml_pl ] s: ( [ gen ] )
                              r: ( [ pl ]
                                   n: ( [ lr ] ) ) ).
Staat:<tree 0> =      ( [ m ] s: ( [ gen ] )
                          n: ( [ pl ] ) ).
Hemd:<tree 0> =       ( [ ] s: ( [ gen ] )
                        n: ( [ pl ] ) ).
Zeit:<tree 0> =       ( [ f ] n: ( [ pl] ) ).
Farbe:<tree 0>=       ( [ f ] n: ( [pl ] ) ).
Sieben:<tree 0> =     ( [ f ] n: ( [pl ] ) ).
```

```
Baer:<tree 0> =      ( [ m ] n: ( [ hr] )
                            n: ( [ pl ] ) ).
Hase:<tree 0> =      ( [ m ] n: ( [ hr] )
                            n: ( [ pl ] ) ).
Fels:<tree 0> =      ( [ m ] n: ( [ hr] s: ( [ gen ] ) )
                            n: ( [ pl ] ) ).
Name:<tree 0> =      ( [ m ] n: ( [hr] s: ( [ gen ] ) )
                            n: ( [ pl ] ) ).
Herz:<tree 0> =      ( [ ] n: ( [ hr ] s: ( [ gen ] ) )
                          n: ( [ pl ] ) ).
```